/*-*- mode:unix-assembly; indent-tabs-mode:t; tab-width:8; coding:utf-8     -*-â
âvi: set et ft=asm ts=8 tw=8 fenc=utf-8                                     :viâ
ââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââ¡
â Copyright 2020 Justine Alexandra Roberts Tunney                              â
â                                                                              â
â Permission to use, copy, modify, and/or distribute this software for         â
â any purpose with or without fee is hereby granted, provided that the         â
â above copyright notice and this permission notice appear in all copies.      â
â                                                                              â
â THE SOFTWARE IS PROVIDED "AS IS" AND THE AUTHOR DISCLAIMS ALL                â
â WARRANTIES WITH REGARD TO THIS SOFTWARE INCLUDING ALL IMPLIED                â
â WARRANTIES OF MERCHANTABILITY AND FITNESS. IN NO EVENT SHALL THE             â
â AUTHOR BE LIABLE FOR ANY SPECIAL, DIRECT, INDIRECT, OR CONSEQUENTIAL         â
â DAMAGES OR ANY DAMAGES WHATSOEVER RESULTING FROM LOSS OF USE, DATA OR        â
â PROFITS, WHETHER IN AN ACTION OF CONTRACT, NEGLIGENCE OR OTHER               â
â TORTIOUS ACTION, ARISING OUT OF OR IN CONNECTION WITH THE USE OR             â
â PERFORMANCE OF THIS SOFTWARE.                                                â
ââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââââ*/
#include "libc/macros.internal.h"
.text.windows

//	Translates function call from code built w/ MS-style compiler.
//
//	This wraps WinMain() and callback functions passed to Win32 API.
//	Please note an intermediary jump slot is needed to set %rax.
//
//	@param	%rax is function address
//	@param	%rcx,%rdx,%r8,%r9
//	@return	%rax,%xmm0
//	@note	slower than __sysv2nt
//	@see	NT2SYSV() macro
__nt2sysv:
	push	%rbp
	mov	%rsp,%rbp
//	TODO(jart): We should probably find some way to use our own
//	            stack when Windows delivers signals ;_;
	.profilable
	sub	$0x100,%rsp
	push	%rbx
	push	%rdi
	push	%rsi
	pushf					# TODO(jart): Do we need it?
	lea	-0x80(%rbp),%rdi
	call	_savexmm
	mov	%rcx,%rdi
	mov	%rdx,%rsi
	mov	%r8,%rdx
	mov	%r9,%rcx
	call	*%rax
	lea	-0x80(%rbp),%rdi
	call	_loadxmm
	popf
	pop	%rsi
	pop	%rdi
	pop	%rbx
	leave
	ret
	.endfn	__nt2sysv,globl,hidden
